(* Asymptotic expected number of hairpins in ALL structures with theta=1 and stickiness p=3/8*)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota−Lalley−Woods Theorem. Grammar with S, T where T means there is at least one base pair. *)

```
Clear["*"]
(* S arrow D or T; T arrow (E) or T* or (T) or S(D1) or S(T); D0 arrow * or D0 *;
D1 arrow *** or D1 *  *)
Clear[p, S, D0, D1, z, eqn0, eqn, F, z0, y0, dFdz0fz0S0, d2Fdy0fz0S0];

eqn0 = {S == z + z S + z^2 S + z^2 S^2}; (* grammar for all str, theta=1,p=1 *)
CellPrint["eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1"]
eqn1 = {S == D0 + T, T == z^2 D1 + z T + z^2 T + z^2 S T + z^2 S D1,
   D0 == z + z D0, D1 == z + z D1};
CellPrint[
 "eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0"]
Eliminate[eqn0, {D0, D1, T}]
Collect[%, S, Simplify]
Eliminate[eqn1, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["We derive the SAME equations
   from eqn0,eqn1 showing the underlying grammar is correct."]
p = 3 / 8;
eqn = {S == D0 + T, T == p z^2 D1 + z T + p z^2 T + p z^2 S T + p z^2 S D1,
   D0 == z + z D0, D1 == z + z D1};
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S = ... to get F"]
F = (3 S^2 z^2 + 8 z) / - (-8 + 8 z + 3 z^2);


NSolve[{F == S, D[F, S] == 1}, {z, S}]
z0 = 0.4899528946607232`;
y0 = 2.3329595143758586`;


dFdz0fz0S0 = D[F, z] /. {z → z0, S → y0}

d2Fdy0fz0S0 = D[F, {S, 2}] /. {z → z0, S → y0}
c = Sqrt[z0 dFdz0fz0S0 / (2 Pi d2Fdy0fz0S0)]
c * (1 / z0) ^n n^ (-3 / 2)
```

eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1

eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S\left(-1 + z + z^2\right) == -z$$

$$S^2 z^2 + S(-1+z+z^2) == -z$$

$$S^2 z^2 + S(-1+z+z^2) == -z$$

We derive the SAME equations from eqn0,eqn1 showing the underlying grammar is correct.

$$3 S^2 z^2 + S(-8+8z+3z^2) == -8z$$

$$3 S^2 z^2 + S(-8+8z+3z^2) == -8z$$

From previous expression, write S = ... to get F

```
{{z → 5.4427, S → -0.699966}, {z → 0.489953, S → 2.33296},
 {z → -0.29966 - 1.60526 i, S → -0.816497 - 0.983019 i},
 {z → -0.29966 + 1.60526 i, S → -0.816497 + 0.983019 i}}
```

14.7377

0.42864

1.6374

$$\frac{1.6374 \times 2.04101^n}{n^{3/2}}$$

(*Now, we compute mean,variance using Drmota's Theorem*)

```
Clear["*"]
p = 3 / 8;
eqn = {S == D0 + T, T == p u z^2 D1 + z T + p z^2 T + p z^2 S T + p u z^2 S D1,
   D0 == z + z D0, D1 == z + z D1};


CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S=... to obtain F"]
F = (3 S^2 (-1 + z) z^2 + z (-8 + 8 z + 3 z^2) - (3 u z^3 + 3 S u z^3)) / - (8 - 16 z + 5 z^2 + 6 z^3);

f = (F /. S → s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]
c = Denominator[Together[s - f]]
dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
d = Denominator[Together[1 - dfs]]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
If[Denominator[Together[s - f]] == Denominator[Together[1 - dfs]],
 CellPrint["Denominator of (s-f) same as that of (1-dfs)"],
 CellPrint["Denominator of (s-f) different than than of (1-dfs)"]]
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]
```

```
(* res =Resultant[s-f,1-dfs,s] *)


(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)


res /. z → z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)




rho = 0.4899528946607232`;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u → 1) /. z[1] → rho
CellPrint[
 "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u → 1) /. z[1] → rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = (((D[dzu, u] /. u → 1) /. z[1] → rho) /. z'[1] → dzuEvaluatedAt1)
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu
CellPrint["Now compute the standard deviation sigma"]
stdev = Sqrt[var]
```

Eliminate all variables except S,u,z

$$8 S - 8 z - 16 S z + 8 z^2 + 5 S z^2 - 3 S^2 z^2 + 3 z^3 + 6 S z^3 + 3 S^2 z^3 == (3 + 3 S) u z^3$$

$$3 S^2 (-1 + z) z^2 + z \left(-8 + 8 z + 3 z^2\right) + S \left(8 - 16 z + 5 z^2 + 6 z^3\right) == 3 u z^3 + 3 S u z^3$$

From previous expression, write S=... to obtain F

$$\frac{3 s^2 (-1 + z) z^2 - 3 u z^3 - 3 s u z^3 + z \left(-8 + 8 z + 3 z^2\right)}{-8 + 16 z - 5 z^2 - 6 z^3}$$

$$s - \frac{3 s^2 (-1 + z) z^2 - 3 u z^3 - 3 s u z^3 + z \left(-8 + 8 z + 3 z^2\right)}{-8 + 16 z - 5 z^2 - 6 z^3}$$

$$\left(8 s - 8 z - 16 s z + 8 z^2 + 5 s z^2 - 3 s^2 z^2 + 3 z^3 + 6 s z^3 + 3 s^2 z^3 - 3 u z^3 - 3 s u z^3\right) / \left(8 - 16 z + 5 z^2 + 6 z^3\right)$$

$$8 s - 8 z - 16 s z + 8 z^2 + 5 s z^2 - 3 s^2 z^2 + 3 z^3 + 6 s z^3 + 3 s^2 z^3 - 3 u z^3 - 3 s u z^3$$

$$8 - 16 z + 5 z^2 + 6 z^3$$

$$\frac{6 s (-1 + z) z^2 - 3 u z^3}{-8 + 16 z - 5 z^2 - 6 z^3}$$

$$1 - \frac{6\, s\, (-1 + z)\, z^2 - 3\, u\, z^3}{-8 + 16\, z - 5\, z^2 - 6\, z^3}$$

$$\frac{8 - 16\, z + 5\, z^2 - 6\, s\, z^2 + 6\, z^3 + 6\, s\, z^3 - 3\, u\, z^3}{8 - 16\, z + 5\, z^2 + 6\, z^3}$$

$$8 - 16\, z + 5\, z^2 - 6\, s\, z^2 + 6\, z^3 + 6\, s\, z^3 - 3\, u\, z^3$$

$$8 - 16\, z + 5\, z^2 + 6\, z^3$$

Denominator of (s-f) same as that of (1-dfs)

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$192\, z^2 - 960\, z^3 + 1776\, z^4 - 1488\, z^5 - 144\, u\, z^5 + 555\, z^6 + 432\, u\, z^6 - 75\, z^7 - 486\, u\, z^7 + 198\, u\, z^8 + 27\, u^2\, z^8 - 27\, u^2\, z^9$$

```
192 z[u]^2 - 960 z[u]^3 + 1776 z[u]^4 - 1488 z[u]^5 - 144 u z[u]^5 + 555 z[u]^6 +
  432 u z[u]^6 - 75 z[u]^7 - 486 u z[u]^7 + 198 u z[u]^8 + 27 u^2 z[u]^8 - 27 u^2 z[u]^9
```

```
-144 z[u]^5 + 432 z[u]^6 - 486 z[u]^7 + 198 z[u]^8 + 54 u z[u]^8 -
  54 u z[u]^9 + 384 z[u] Derivative[1][z][u] - 2880 z[u]^2 Derivative[1][z][u] +
  7104 z[u]^3 Derivative[1][z][u] - 7440 z[u]^4 Derivative[1][z][u] -
  720 u z[u]^4 Derivative[1][z][u] + 3330 z[u]^5 Derivative[1][z][u] +
  2592 u z[u]^5 Derivative[1][z][u] - 525 z[u]^6 Derivative[1][z][u] -
  3402 u z[u]^6 Derivative[1][z][u] + 1584 u z[u]^7 Derivative[1][z][u] +
  216 u^2 z[u]^7 Derivative[1][z][u] - 243 u^2 z[u]^8 Derivative[1][z][u]
```

```
-144 z[u]^5 + 432 z[u]^6 - 486 z[u]^7 + 198 z[u]^8 + 54 u z[u]^8 - 54 u z[u]^9 -
  3 z[u] (-128 + 960 z[u] - 2368 z[u]^2 + 80 (31 + 3 u) z[u]^3 - 6 (185 + 144 u) z[u]^4 +
     7 (25 + 162 u) z[u]^5 - 24 u (22 + 3 u) z[u]^6 + 81 u^2 z[u]^7) Derivative[1][z][u]
```

```
{{Derivative[1][z][u] -> -(6 (8 z[u]^4 - 24 z[u]^5 + 27 z[u]^6 - 11 z[u]^7 - 3 u z[u]^7 + 3 u z[u]^8)) /
      (-128 + 960 z[u] - 2368 z[u]^2 + 2480 z[u]^3 + 240 u z[u]^3 - 1110 z[u]^4 -
         864 u z[u]^4 + 175 z[u]^5 + 1134 u z[u]^5 - 528 u z[u]^6 - 72 u^2 z[u]^6 + 81 u^2 z[u]^7)}}
```

-0.0465593

According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next.

0.0950281

0.0359223

Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu

0.0307406

Now compute the standard deviation sigma

0.17533